In[1]:=
```
(* v contains variable names of the chart *)
v = {t, r, θ, ϕ};
(* g is the metric g_μν *)
g = DiagonalMatrix[{-1, a[t]^2/(1 - k r^2), a[t]^2 r^2, a[t]^2 r^2 Sin[θ]^2}];
(* gInv is the inverse metric g^μν *)
gInv = Inverse[g];
```

In[2]:=
```
(* Calculating Γ^ρ_μν = Γ[ρ,μ,ν] by formular (3.1 .30) of Wald *)
Γ[ρ_, μ_, ν_] := Γ[ρ, μ, ν] =
  FullSimplify[1/2 Sum[gInv[[ρ, σ]] (∂_v[[μ]] g[[ν, σ]] + ∂_v[[ν]] g[[μ, σ]] - ∂_v[[σ]] g[[μ, ν]]), {σ, 1, 4}]];

(* Calculating the Ricci tensor R_μρ = R[μ,ρ] by (3.4 .5) of Wald *)
R[μ_, ρ_] := R[μ, ρ] = FullSimplify[Sum[∂_v[[ν]] Γ[ν, μ, ρ], {ν, 1, 4}] -
    ∂_v[[μ]] (Sum[Γ[ν, ν, ρ], {ν, 1, 4}]) + Sum[(Sum[(Γ[α, μ, ρ] Γ[ν, α, ν] - Γ[α, ν, ρ] Γ[ν, α, μ]), {ν, 1, 4}]), {α, 1, 4}]];

(* Calculate the Ricci scalar RR = R_μν g^νμ *)
RR = FullSimplify[Sum[(Sum[R[μ, ν] gInv[[ν, μ]], {ν, 1, 4}]), {μ, 1, 4}]];

(* Finally, calculate G_μν = R_μν - 1/2 RR g_μν *)
G[μ_, ν_] := G[μ, ν] = FullSimplify[R[μ, ν] - 1/2 RR g[[μ, ν]]];
```

In[3]:=
```
(* Display all the Γ's, one table for each upper index, starting with v[[1]] *)
```

```
In[4]:= Table[MatrixForm[Table[Γ[m, i, j], {i, 1, 4}, {j, 1, 4}]], {m, 1, 4}] // TableForm
```

Out[4]//TableForm=

$$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & \frac{a[t]\, a'[t]}{1-k\, r^2} & 0 & 0 \\ 0 & 0 & r^2\, a[t]\, a'[t] & 0 \\ 0 & 0 & 0 & r^2\, a[t]\, \mathrm{Sin}[\theta]^2\, a'[t] \end{pmatrix}$$

$$\begin{pmatrix} 0 & \frac{a'[t]}{a[t]} & 0 & 0 \\ \frac{a'[t]}{a[t]} & \frac{k\, r}{1-k\, r^2} & 0 & 0 \\ 0 & 0 & r\,(-1+k\, r^2) & 0 \\ 0 & 0 & 0 & r\,(-1+k\, r^2)\, \mathrm{Sin}[\theta]^2 \end{pmatrix}$$

$$\begin{pmatrix} 0 & 0 & \frac{a'[t]}{a[t]} & 0 \\ 0 & 0 & \frac{1}{r} & 0 \\ \frac{a'[t]}{a[t]} & \frac{1}{r} & 0 & 0 \\ 0 & 0 & 0 & -\mathrm{Cos}[\theta]\, \mathrm{Sin}[\theta] \end{pmatrix}$$

$$\begin{pmatrix} 0 & 0 & 0 & \frac{a'[t]}{a[t]} \\ 0 & 0 & 0 & \frac{1}{r} \\ 0 & 0 & 0 & \mathrm{Cot}[\theta] \\ \frac{a'[t]}{a[t]} & \frac{1}{r} & \mathrm{Cot}[\theta] & 0 \end{pmatrix}$$

```
(* Display Ricci tensor as a 4*4 matrix *)

In[5]:= Table[R[i, j], {i, 1, 4}, {j, 1, 4}] // MatrixForm
```

Out[5]//MatrixForm=

$$\begin{pmatrix} -\frac{3\, a''[t]}{a[t]} & 0 & 0 & 0 \\ 0 & \frac{2\,(k+a'[t]^2)+a[t]\, a''[t]}{1-k\, r^2} & 0 & 0 \\ 0 & 0 & r^2\,(2\,(k+a'[t]^2)+a[t]\, a''[t]) & 0 \\ 0 & 0 & 0 & r^2\, \mathrm{Sin}[\theta]^2\,(2\,(k+a'[t]^2)+a[t]\, a''[t]) \end{pmatrix}$$

```
(* Ricci curvature *)

In[6]:= RR
```

Out[6]= $$\frac{6\,(k+a'[t]^2+a[t]\, a''[t])}{a[t]^2}$$

```
In[7]:= (* And finally, Einstein' s tensor *)
        Table[G[i, j], {i, 1, 4}, {j, 1, 4}] // MatrixForm
```

Out[7]//MatrixForm=

$$\begin{pmatrix} \frac{3\,(k+a'[t]^2)}{a[t]^2} & 0 & 0 & 0 \\ 0 & \frac{k+a'[t]^2+2\, a[t]\, a''[t]}{-1+k\, r^2} & 0 & 0 \\ 0 & 0 & -r^2\,(k+a'[t]^2+2\, a[t]\, a''[t]) & 0 \\ 0 & 0 & 0 & -r^2\, \mathrm{Sin}[\theta]^2\,(k+a'[t]^2+2\, a[t]\, a''[t]) \end{pmatrix}$$